

Supplemental Models for Example 1 in Mplus

```

TITLE: Model 3 via MLM using Unbalanced Time:
          Attitudes Predict Intercept, Linear, and Quadratic Age Slopes
DATA: FILE = Example1.csv;    ! Syntax in same folder as data
VARIABLE:
  ! List of variables in data file
  NAMES = PersonID Att12 occasion age risky mon roundage
         time att4 timesq mon3;
  ! Variables to be analyzed in this model
  USEVARIABLE = time timesq att4 risky;
  MISSING ARE ALL (-999);    ! Missing data identifier
  ! MLM options
  CLUSTER = PersonID;        ! Level-2 ID
  BETWEEN = att4;            ! Observed ONLY level-2 predictors
  WITHIN = time timesq;      ! Observed ONLY level-1 predictors

ANALYSIS: TYPE = TWOLEVEL RANDOM; ESTIMATOR = ML;

MODEL: ! LEVEL-1 = WITHIN, LEVEL-2 = BETWEEN
%WITHIN%
  risky;                      ! L1 R: Residual variance
  lin | risky ON time;         ! Placeholder for linear age slope
  quad | risky ON timesq       ! Placeholder for quadratic age slope

%BETWEEN%
[risky lin quad];             ! Fixed intercept, linear, quad slopes
risky lin quad@0;             ! L2 G: Random effects variances (quad=0)
risky WITH lin;               ! L2 G: Random effects covariance
risky lin quad ON att4;       ! Att-> risky int, linear, quad slopes

```

MODEL FIT INFORMATION

Number of Free Parameters	10
Loglikelihood	
H0 Value	-3799.718
Information Criteria	
Akaike (AIC)	7619.436
Bayesian (BIC)	7671.878
Sample-Size Adjusted BIC	7640.112
(n* = (n + 2) / 24)	

MODEL RESULTS

		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level					
Residual Variances → Level-1 residual variance					
RISKY		8.324	0.372	22.353	0.000
Between Level					
LIN ON					
ATT4		-0.900	0.242	-3.721	0.000 gamma11
QUAD ON					
ATT4		-0.064	0.036	-1.761	0.078 gamma21
RISKY ON					
ATT4		-3.476	0.580	-5.989	0.000 gamma01
RISKY WITH → Level-2 BP random effect covariance					
LIN		1.884	0.356	5.286	0.000
Intercepts					
RISKY		23.299	0.350	66.573	0.000 gamma00
LIN		1.948	0.146	13.336	0.000 gamma10
QUAD		0.142	0.022	6.462	0.000 gamma20
Residual Variances → Level-2 BP random effect variances remaining					
RISKY		18.083	2.204	8.206	0.000
LIN		0.486	0.080	6.087	0.000
QUAD		0.000	0.000	999.000	999.000

```

TITLE: Model 3 via SEM using Unbalanced Time:
          Attitudes Predict Intercept, Linear, and Quadratic Age Slopes
DATA: FILE = Example1.csv;    ! Syntax in same folder as data

! Unstacking to multivariate format
DATA LONGTOWIDE:
! Names of old stacked former variables (without numbers)
LONG = risky|time;
! Names of new multivariate variables (that use numbers)
WIDE = risky12-risky18|age12-age18;
! Variable with level-2 ID info
IDVARIABLE = PersonID;
! Old level-1 identifier
REPETITION = roundage (12 13 14 15 16 17 18);

VARIABLE:
! List of variables in original data file
NAMES = PersonID Att12 occasion age risky mon roundage
       time att4 timesq mon3;
! Variables to be analyzed in this model
USEVARIABLE = att4 risky12-risky18 age12-age18;
MISSING ARE ALL (-999);    ! Missing data identifier
TSCORES = age12-age18;    ! Exact time indicator

ANALYSIS: TYPE = RANDOM; ESTIMATOR = ML;
MODEL:
[risky12-risky18@0];      ! All variable intercepts fixed to 0
risky12-risky18 (Resvar); ! L1 R residual variances held equal

! Risky behavior quadratic growth model using exact age as loadings
Int Lin Quad | risky12-risky18 AT age12-age18;

! Fixed intercept, linear, quad slopes
[Int Lin Quad];
! L2 G: Random effects variances (quad=0)
Int Lin Quad@0;
! L2 G: Random effects covariance
Int WITH Lin;
! Attitudes --> risky int, linear, quad slopes
Int Lin Quad ON att4;

```

MODEL FIT INFORMATION

Number of Free Parameters		10			
Loglikelihood					
H0 Value		-3799.718			
Information Criteria					
Akaike (AIC)		7619.437			
Bayesian (BIC)		7652.419			
Sample-Size Adjusted BIC		7620.738			
(n* = (n + 2) / 24)					
		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
INT	ON				
ATT4		-3.476	0.580	-5.989	0.000 gamma01
LIN	ON				
ATT4		-0.900	0.242	-3.721	0.000 gamma11
QUAD	ON				
ATT4		-0.064	0.036	-1.761	0.078 gamma21
INT	WITH → L2 BP random effect covariance				
LIN		1.884	0.356	5.286	0.000
Intercepts					
RISKY12		0.000	0.000	999.000	999.000
RISKY13		0.000	0.000	999.000	999.000
RISKY14		0.000	0.000	999.000	999.000
RISKY15		0.000	0.000	999.000	999.000
RISKY16		0.000	0.000	999.000	999.000
RISKY17		0.000	0.000	999.000	999.000
RISKY18		0.000	0.000	999.000	999.000
INT		23.299	0.350	66.571	0.000 gamma00
LIN		1.948	0.146	13.336	0.000 gamma10
QUAD		0.142	0.022	6.462	0.000 gamma20
Residual Variances					
RISKY12		8.324	0.372	22.353	0.000 L1 res var
RISKY13		8.324	0.372	22.353	0.000
RISKY14		8.324	0.372	22.353	0.000
RISKY15		8.324	0.372	22.353	0.000
RISKY16		8.324	0.372	22.353	0.000
RISKY17		8.324	0.372	22.353	0.000
RISKY18		8.324	0.372	22.353	0.000
INT		18.084	2.204	8.206	0.000 L2 BP int var
LIN		0.486	0.080	6.087	0.000 L2 BP lin var
QUAD		0.000	0.000	999.000	999.000 L2 BP quad var